ABSTRACT

There are many algorithms which are able to find a function root. Like well-known Newton method, Brent’s numerical root-finding method ... etc. But all these algorithms are calculus-based techniques, and need object function was continuous or differential, and need a good start point to search. Whereas, Genetic Algorithm doesn’t need to consider these problems, just give it a pretty wide searching interval and see if the value of object function is near zero. This paper is trying to use Genetic Algorithm to find out a function root and displays its power.

INTRODUCTION

The Genetic Algorithms are under SAS/IML® module. Genetic Algorithms are very powerful. GAs are a family of searching algorithms that could be applied to solve any optimization problems (find out a function root also could be corresponding to a optimization problem). GAs principle are more like natural selection and evolution (i.e. survival of the fitness theory). GAs are especially useful for the objective function which is not differentiable or continuous. GAs require more computation resource and time than other calculus-based techniques. GAs are not random searching algorithms due to its ability of keeping the gene of best members into next generation. That means it have a pretty much convergent speed, especially for real-valued encoding.

GAs have the following basic steps (it is taken from IML documentation):

1) Encoding: The general structure and form of the solution.
2) Objective: The function to be optimized. IML also enables you to specify whether the function is to be minimized or maximized.
3) Selection: How members of the current solution population will be chosen to be parents to propagate the next generation.
4) Crossover: How the attributes of parent solutions will be combined to produce new offspring solutions.
5) Mutation: How random variation will be introduced into the new offspring solutions to maintain genetic diversity.
6) Repeated from 2): When reach the max iterative number, and exit GA.

WHAT IS A FUNCTION ROOT

Suppose we have a function, we put everything at left side of equal symbol, and put zero at the right side of equal symbol, if we can find a solution to satisfy that form function, and then we can say we find a root for that function. Take for example: we have a function like \( x^2 = 4 \) and transform it like \( x^2 - 4 = 0 \) (i.e. \( f(x) = 0 \)). If we can find a \( x_0 \) to satisfy \( f(x_0) = 0 \), then we can say \( x_0 \) is a root of \( f(x) \). For this case, \( x_0 = 2 \) and \( x_0 = -2 \) both are the root of \( f(x) \).

WHY TO USE GENETIC ALGORITHM

In SAS/IML, there are two functions FROOT() and POLYROOT () which can be used to find out a function’s root. Why would we need Genetic Algorithm to solve such problem? The reasons are:

1) GA can handle the function which is not differentiable or continuous, whereas FROOT() can’t handle such function.

2) FROOT() must offer a good start point to search root. E.g. you must specify a special interval \([a, b]\) which could make \( f(a) * f(b) < 0 \). whereas GA don’t need to consider about it, just specify a very wide
interval( which would contain a root ) like \([-10, 10]\), and see whether the object function value is near zero (every small like 1e-6).

3) If you have a hundred, a thousand of functions need to look for a root. FROOT() almost couldn’t accomplish such task due to need to specify a good start point.

4) POLYROOT() is only applied to find the roots of a real polynomial.

**EXAMPLE**

Suppose \(f(x) = (0.3-x)^2 / (1+x)^2 - 48\). It is obvious that \(x_0=-1\) is not a differential point (i.e. \((1+x)^2 = 0\)). You cannot use FROOT() for this case unless you define a good interval \([a, b]\) which could make \(f(a)f(b) < 0\). Let’s see what it looks like:

```plaintext
proc iml;
/* Define a object function */
start func(x);
  v= (0.3-x)**2/(1+x)**2 - 48;
  return (v);
finish;

x= do(-3,-1.11,0.01)||{.}||do(-0.9,1,0.01);
y=func(x);
call series x=x y=y other="refline 0/axis=y";
quit;
```

![Graph of f(x) = (0.3-x)^2 / (1+x)^2 - 48](image)

**Figure 1.** \(f(x) = (0.3-x)^2 / (1+x)^2 - 48\)

Let’s see how to use FROOT() to find a root:
proc iml;
/* Define a object function */
start func(x);
  v= (0.3-x)#2/(1+x)#2 - 48 ;
  return (v);
finish;
bounds = {-1.5, -1.1,
          -0.9, -0.5};
fbounds = Func(bounds[,1]) || Func(bounds[,2]);
roots = froot("Func", bounds);
print roots;
quit;

roots
-1.254501
-0.787165

Output 1. Output from IML Statement of FROOT()

Let's see how to use GA to find a root.

Finding a root of function could be corresponding to a optimization problem, just add absolute value on f(x). So the optimization problem of finding a function root could be expressed as :

\[ \text{MIN } |f(x)| \]

As long as \(|f(x)|\) is near zero , we can say we find a root of function. As you can see the value of objection function is 7.105E-15 , which means it is very small( near zero ), so we can say we found a function root -1.254501 that is the same as function FROOT() does. You also notice that we just define a very wide range [-10,10] ,not like FROOT() need to consider \(f(a)\times f(b) < 0\) and consider if there is a uncontinuous point in the searching interval. Of course, we can define another interval to get another root. This example is just showed how easy to use for GA and how powerful for GA. More details check the annotation in the following code.

proc iml;
/*Define a object function. NOTE: Add an ABS()on f(x)*/
start func(x);
  v=abs( (0.3-x)#2/(1+x)#2 - 48 );
  return (v);
finish;

/*The first 1 means fixed-length real encoding. since there is only one
parameter X (fixed-length) in object function and searching interval is 
continuous for X (real)/
/*The second 1 means the size of encoding. since there is only one
parameter X in object function */
/* 1234 is initial random seed*/
id=gasetup(1,1,1234);

/*id is what returned from gasetup()*/
/* 0 means minimize user module*/
/* “func” is user module we define above*/
call gasetobj(id,0,"func");

/* 100 is selecting the best 100 members into next generation*/
/* 1 is dual tournament selective method*/
/* 0.95 is selective pressure. The member with the best objective value is chosen with a Probability 0.95 */
call gasetsel(id,100,1,0.95);

/* 0.05 is crossover probability. For real encoding I recommend to use a small value to avoid to leak out the best member*/
/* 4 define a kind of crossover method – heurisitic */
call gasetcro(id,0.05,4);

/* 0.05 is mutation probability. Use the default mutation method – uniform*/
call gasetmut(id,0.05);

/*Initial population size is 10000 members. Range is [-10,10]*/
call gainit(id,10000,{-10,10});
niter = 100; /* The number of iterations*/
summary = j(niter,2);
matttrib summary [c = {"Min Value", "Avg Value"} l=""];
do i = 1 to niter;
   /*Get the next generation*/
call garegen(id);
   /*Get the value of Objection function */
call gagetval(value, id);
   /*Get the best value of Objection function of this generation*/
   summary[i,1] = value[1];
   /*Get the average value of Objection function of this generation*/
   summary[i,2] = value[:];
end;
/*Get the best member X and the best value of object function*/
call gagetmem(mem, value, id, 1);
print mem[ l="ROOT:"],
   "Min Value: " value[1 = ""
iteration = t(1:niter);
print iteration summary;
call gaend(id);
quit;

ROOT:
-1.254501
Min Value: 7.105E-15

<table>
<thead>
<tr>
<th>Iteration</th>
<th>MinValue</th>
<th>AvgValue</th>
</tr>
</thead>
<tbody>
<tr>
<td>95</td>
<td>7.105E-15</td>
<td>12.584355</td>
</tr>
<tr>
<td>96</td>
<td>7.105E-15</td>
<td>3.1274638</td>
</tr>
<tr>
<td>97</td>
<td>7.105E-15</td>
<td>3.1356642</td>
</tr>
<tr>
<td>98</td>
<td>7.105E-15</td>
<td>3.379379</td>
</tr>
<tr>
<td>99</td>
<td>7.105E-15</td>
<td>4.092118</td>
</tr>
<tr>
<td>100</td>
<td>7.105E-15</td>
<td>3.2699371</td>
</tr>
</tbody>
</table>

Output 2. Output from IML Statement of GA
CONCLUSION
Beyond traditional calculus-based techniques, the Genetic Algorithms have itself qualities for the problem of looking for a function root. When you meet a problem that traditional method can’t solve, consider using Genetic Algorithms.

REFERENCES

CONTACT INFORMATION
Your comments and questions are valued and encouraged. Contact the author at:
   Xia Ke Shan
   iFRE Inc.
   Beijing, China
   Phone: +8613521225927
   E-mail: 12135835@qq.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.